Paper 332-2011

Using SAS® PROC TCALIS for Multigroup Structural Equation Modeling with Mean Structures

Fei Gu, University of Kansas, Lawrence, KS Wei Wu, University of Kansas, Lawrence, KS

ABSTRACT

Multigroup structural equation modeling (SEM) is a frequently used technique to evaluate measurement invariance in social and behavioral science research. Before the 9.2 version, SAS was incapable of handling multigroup SEM, but this limitation is resolved in PROC TCALIS in SAS 9.2. For the purpose of illustration, this article provides step-by-step guide to programming the tests of measurement invariance and partial invariance using PROC TCALIS for multigroup SEM with mean structures. Fit indices and parameter estimates are validated, thus providing an alternative tool for researchers who conduct both applied and simulated studies. Other new features (e.g., different types of modeling languages and estimation methods) and limitations (e.g., ordered-categorical SEM and multilevel SEM) of the TCALIS procedure are also briefly mentioned.

INTRODUCTION

Multigroup structural equation modeling (SEM) is a frequently used technique to evaluate measurement invariance in social and behavioral social science research. In the past decades, a variety of commercial software packages have been developed for SEM, including EQS (Bentler & Wu, 2002), Mplus (Muthén & Muthén, 1998-2007), LISREL (Jöreskog & Sörbom, 1996), Mx (Neale, Boker, Xie, & Maes, 2003), and AMOS (Arbuckle, 2003). Since version 8, SAS has also added a procedure into the SAS/STAT product to accommodate SEM (i.e., PROC CALIS). However, the CALIS procedure has one major limitation—its inability of handling multigroup SEM (Fan & Fan, 2005). Although some researchers tried to "trick" SAS to analyze multigroup models (provided that each group had the same sample size), this trick is not generalizable to situations where unequal sample sizes are mostly encountered (Jones-Farmer, Pitts, & Rainer, 2008, Marcoulides & Hershberger, 1997). In addition, using such a trick may give incorrect degrees of freedom. Thus, one must be cautious about using PROC CALIS for multigroup analyses.

Because of such a limitation, SAS is not the first choice to implement multigroup invariance tests to some researchers (e.g., Jones-Farmer et al., 2008). According to Byrne (2004), most literature addressing multigroup invariance has used either LISREL or EQS. However, there might be times when, by convenience or necessity, SAS would be preferred. For example, SAS offers quantitative researchers an extremely flexible environment to conduct various Monte Carlo simulation studies (Fan, Felsövályi, Sivo, & Keenan, 2003). Data simulation and subsequent analyses of the simulated results can be easily implemented in SAS by using a wide variety of descriptive and/or advanced statistical procedures (e.g., PROC MEANS and PROC GLM). Because of the previous limitation in multigroup analyses, simulation studies using PROC CALIS only involved single-group analyses (see Fan & Sivo, 2005; Yang & Green, 2010). Simulation studies involving multigroup SEM can become very laborious if one needs to simulate data in SAS, export the data to another software package, say LISREL, for multigroup analyses, and then, import the output back into SAS for later analyses (e.g., Fan & Sivo, 2009). Generally, a practice that involves data exchange among different software packages is inconvenient and time-consuming in simulation studies.

In SAS 9.2, an experimental procedure, PROC TCALIS, was introduced. The TCALIS procedure is modified with changes and enhancements from the old CALIS procedure. According to the SAS document (SAS Institute, 2008), PROC TCALIS is not a simple functional enhancement of PROC CALIS. The basic computational architecture of PROC TCALIS is quite different from that of PROC CALIS. New features include, but are not limited to, new modeling languages, multigroup analysis, and improved mean structures analysis. With the TCALIS procedure available, the inconvenient data exchange between SAS and other SEM packages in simulation studies can be solved.

Nevertheless, up to this date, no journal article or textbook has provided example program to illustrate the TCALIS procedure for multigroup SEM. Therefore, the purpose of this article is to provide a step-by-step tutorial using PROC TCALIS. Specifically, tests of invariance and partial invariance of mean and covariance structures between two groups are illustrated with an example. We believe that researchers who conduct both applied and simulated studies can benefit from such an alternative tool in their future work.

MEASUREMENT INVARIANCE

SEM models are used to describe the relationships between manifest and/or latent variables. When a particular theoretical model is justified as a good enough approximation to the sample data for a homogenous group, the research question whether the same model holds across heterogeneous groups may come to the interest. Such groups may be defined by any categorical variables in practice (e.g., gender, race/ethnicity, social-economic status, etc.). Testing measurement invariance in the multigroup framework is becoming increasingly popular to answer such related questions. Technically, measurement invariance can be tested at different levels. Detailed discussion of measurement invariance can be found in the literature elsewhere (e.g., Bollen, 1989; Byrne, Shavelson, & Muthén, 1989; Cheung & Rensvold, 2002; Little, 1997; Meredith, 1993; Vandenberg & Lance, 2000), and a brief summary is provided below.

Typically, the first level of invariance is a model with no constraint imposed on any parameter across groups (configural invariance). If the configural model adequately fits the data, then, equivalence of all factor loadings are placed across groups (weak factorial invariance). The weak invariance model is evaluated with the configural invariance model. If model fit between the two nested models are not statistically different, then, equivalence of covariance and/or mean structures can be placed across groups. Otherwise, constraints on factor loadings that caused the lack of fit should be removed, not simultaneously but one at a time, until partial invariance of factor loadings is established. Once the (partial) weak invariance model is satisfied, researchers can examine structures of the mean, the covariance, or both, depending on research questions. Difference of mean structures across groups can be explored by invariance tests on intercepts for observed measures and factors. Alternatively, if covariance structures are of interest, invariance tests of the factor covariance matrix, with or without involving the mean structures, can be conducted (strong factorial invariance). Lastly, invariance of covariance structures of measurement errors, with or without involving the mean structures, should be examined based on the strong invariance model (strict factorial invariance). The invariance tests illustrated in this example only involve factor loadings and mean structures for observed measures and factors, which correspond to the 10th model from the taxonomy of 13 partially nested models operationalized by Marsh et al. (2009, Table 1, p.443).




Figure 1. Path diagram with the structured means in Group 1.

EXAMPLE

DATA

This example data from the book chapter by Thompson and Green (2006, p.139, Table 5.2, Dataset 2) were borrowed, which contain six measured variables aiming to assess preschool children academic (V1-V3) and social school readiness (V4-V6). Preschool children were divided into two groups: Group 1—day-care and Group 2—homecare. In the book chapter, three means and covariance matrices were provided, two for the separate groups and the other for their combined group. What we need here are the two means and covariances matrices for Group 1 and Group 2. Sample sizes are 250 and 150 for Group 1 and Group 2, respectively. In each group, there are two correlated factors (F1 and F2). F1 has the first three indicators, V1-V3; and F2 has the last three, V4-V6. To create the datasets in SAS, different from the raw data collected in rows in applied research, users need to explicitly specify the data type as covariance matrix by adding the dataset option in parentheses, type=cov, in the DATA step (Table 1). Though only two groups are illustrated, it is very easy to generalize the procedures to cases with more than two groups.

Table 1. Creating two separate datasets in the type of covariance matrix in the SAS system

```
data group1(type=cov);
infile datalines missover;
input NAME $ TYPE $ V1-V6;
datalines;
  MEAN 49.14 82.60 104.95 78.58 54.95 119.91
       154.54
V1
  COV
   COV 44.75 90.23
V2
V3 COV 40.98 22.77
                     78.76
V4 COV 41.35
              2.83
                     7.92 220.12
V5
  COV 23.28
              9.12
                     0.75 61.46 159.88
V6
  COV 47.08 22.88 16.05 125.08 84.31 332.26
data group2(type=cov);
infile datalines missover;
input NAME $ TYPE $ V1-V6;
datalines;
  MEAN 55.01 81.22 97.01 72.67 46.57 124.28
V1
   COV 124.93
   COV 52.19 80.67
V2
       64.45 42.85 83.56
V3 COV
       59.95 33.10 38.34 290.65
V4 COV
V5 COV 32.32 16.09 18.29 124.71 169.17
   COV 87.15 39.70 51.82 174.20 108.39 355.22
V6
run;
```

STEPWISE ANALYSIS

The stepwise procedure suggested by Thompson and Green (2006, Table 5.3, pp.141-143) is used to examine difference in factor means under partial invariance so that the selected model fit indices, i.e., chi-square, standardized root mean square residual (SRMR), and root mean square error of approximation (RMSEA), reported from PROC TCALIS can be validated. The metric in this example is defined by fixing the variance of factors to 1 in Group 1 and imposing between-group equality constraints of factor loadings (except in Step 1), and by doing so, we are able to evaluate all between-group constraints on factor loadings. Decision rules between steps are such that, except for Step 1, the chi-square difference test, together with SRMS and RMSEA, is used to assess differential fit of nested models; and the decision of removing individual between-group constraint is based on the modification indices (MI), also known as the Lagrange multiplier (LM) tests, in order to improve the model fit (Chou & Bentler, 1990).

The chi-square difference test is not considered as the best practice by some researchers when mean structures get involved because it is strongly influenced by sample size, and alternative criteria were proposed recently (Chen, 2007; Cheung & Rensvold, 2002; Fan & Sivo, 2009). Consequently, different researchers can reach different decisions. As a software tutorial, we follow the same steps presented in the book chapter (Thompson & Green, 2006, Table 5.4, pp.144-147). Details of these alternative tests for nested models evaluation are not discussed here, but readers can find related information from the articles mentioned above or literature elsewhere. A brief summary of the steps is listed below:

- In Step 1, the same factor model (Figure 1) is evaluated in each group and their combined group (configural invariance). Because the model fits the data adequately, we proceed to evaluate the invariance of factor loadings.
- In Step 2a, all factor loadings are constrained to be equal across groups (weak factorial invariance). Because the chi-square difference test is significant between the configural model and the weak invariance model, one or more constraints needs to be removed to establish the partial invariance on factor loadings.
- In Step 2b, equality constraints that caused the lack of fit are identified by the LM/MI tests, and multiple nested
 models are evaluated by the chi-square difference test, ΔSRMR, and ΔRMSEA. In the final model retained in
 Step 2b, constraints on loadings for V3 and V6 are removed.
- In Step 3a, all intercepts for measures are constrained to be equal across groups, except for those for V3 and V6 because the factor loadings for V3 and V6 are not constrained in the previous model. Intercepts for factors in Group 1 (means for F1 and F2) are arbitrarily set to 0, and they are freely estimated in Group 2. Due to the lack of fit in this step, we proceed to evaluate the partial invariance of intercepts for measures.
- In Step 3b Substep 1, the same tests and decision guidelines are used as those in Step 2b. The final model retained in Step 3b is the one with additional constraints on intercepts for V2 and V5, compared to the one in Step 2b. Then, we proceed to evaluate the invariance of intercepts for factors.
- In Step 4a, intercepts for both factors are constrained to be equal across groups. Due to the lack of fit in this step, we proceed to evaluate the partial invariance of intercepts for factors.
- In Step 4b, constraints on intercepts for F1 and F2 are examined, separately. Based on the same decision guidelines, we conclude that the between-group difference of means for F1 is not statistically significant, but they are different for F2.

SAS PROGRAMMING

In the TCALIS procedure, there are a collection of modeling languages available to specify models (e.g., LINEQS, LISMOD, and PATH), each reflecting different modeling terminology and philosophies. Traditionally, most users chose to use the LINEQS language in the old CALIS procedure, and it is still available in PROC TCALIS. Also available is the LISMOD language, which stands for LISrel MODeling. It meets the need of users who are familiar with LISREL models (Jöreskog & Sörbom, 1996). The PATH language seems to be the most flexible, and the following programs are written with the PATH language. Other different modeling languages are also provided (e.g., FACTOR and MSTRUCT), but they are mostly designed for their specific purposes, thus omitted in this illustration. Essentially, different modeling languages are different ways of doing the same analyses.

For simplicity, a model specified by using the PATH language is called a PATH model. A complete model specification usually needs three major statements (i.e., PATH, PVAR, and PCOV) in a PATH model. When mean structures are involved, the MEAN statement is also used.

By using Convention 1 (McDonald & Ho, 2002) to draw the path diagram, a direct translation from the figure to the PATH model is very convenient by following the rules below:

- Each single-headed arrow in the path diagram is specified in the PATH statement.
- Each double-headed arrow pointing to a single variable is specified in the PVAR statement.
- Each double-headed arrow pointing to two different variables is specified in the PCOV statement.

The TCALIS procedure is invoked by the PROC TCALIS statement, and dataset information and estimation method are specified by using options in this statement. The DATA= option tells SAS which dataset is used. For example, in Table 2, the single-group model for Group 1 uses the SAS dataset "group1", and because the type of this dataset is covariance, the COV and NOBS= options are needed, that is, PROC TCALIS will analyze a covariance matrix instead of a raw dataset, and it is based on 250 observations. The METHOD= option specify the estimation method. Normal-theory maximum-likelihood method is used throughout this example.

In the PATH statement, all single-headed arrows are translated to the corresponding path entries. An arrow symbol, -> (or <-), is used between two variables. If A is the antecedent variable to B, then the order of A and B in the path entry is interchangeable as long as the head of the arrow points to the outcome variable, B. In other words, A -> B is equivalent to B <- A. A fixed value or a unique parameter name is also specified to each path entry. Fixed values represent fixed model parameters; whereas a parameter name tells SAS that this particular parameter will be freely estimated in the model. Path entries are separated by a comma. For example, in Table 2, the factor loadings from F1 to V1, V2, and V3 are specified as b1, b2, and b3, respectively; and those from F2 to V4, V5, and V6 are b4, b5, and b6, respectively, in both single-group models for Group 1 and Group2. The multigroup model in Table 2 specifies the

same path entries in Group 1 as those in single-group models. Path specification for Group 2 in multigroup model will be discussed later.

In the PVAR statement, all double-headed arrows pointing to a single variable, representing variance parameters for factor disturbances and measurement errors, are translated to corresponding variances. One or multiple variables are listed on the left side of an equality sign, and one or multiple fixed values and/or unique parameter names are listed on the right side of an equality sign. Similarly, a fixed value represents a fixed parameter; whereas a parameter name represents a freely estimated parameter. Variance entries are separated by a comma. For example, in Table 2, three variance entries are specified in single-group models for Group 1 and Group 2, that is, variances of F1 and F2 are set to 1 in order to define the metric, and six error variances are defined as freely estimated parameters, labeled as e1 to e6 associated with measured variables V1 toV6, separately. The syntax V1-V6 is a handy shortcut to represent the list from V1 to V6. So does e1-e6. Users can always type all six entries (i.e., V1=e1, V2=e2, ..., V6=e6) to achieve the same thing. The multigroup model in Table 2 specifies the same variance entries in Group 1 as those in single-group models. Variance specification for Group 2 in multigroup model is also going to be discussed later.

In the PCOV statement, all double-headed arrows pointing to two different variables, representing covariance parameters, are translated to the corresponding covariances. Two variables separated by a space are listed on the left side of an equality sign, and a fixed value or a unique parameter name is listed on the right side of an equality sign. The usage of fixed values and parameter names is the same as in the PATH and PVAR statement. Different covariance entries are separated by a comma. For example, in Table 2, there is only one pair of correlated disturbances in each model, thus one covariance entry specified, in both single-group models for Group 1 and Group 2, that is, the covariance is a freely estimated parameter, d12, in the model. The multigroup model in Table 2 specifies the same covariance entry in Group 1 as the single-group models. Covariance specification, together with the path and variance specification, for Group 2 in multigroup model is discussed in the next paragraph.

Now, we come to explain the model specification of Group 2 in multigroup SEM in Table 2. First, there are two GROUP statements, each designating a group number. Here, Group 1 and Group 2 are defined. The DATA= and NOBS= options in the GROUP statement are self-explanatory, that is, Group 1 uses the SAS dataset "group1", and it has 250 observations; whereas Group 2 uses "group2", and 150 observations. Second, there are two MODEL statements, each signifying the beginning of a model specification block and designating a model number for the model. Until another MODEL statement is encountered, the PATH, PVAR, and PCOV statement in between belongs to the precedent model specification block. The GROUP= option in the MODEL statement can list one or multiple integers which represent the groups to be fitted by the model. The statement (model 5 / group=6 7;) means that model specification 5 fits Group 6 and 7 (if defined in previous GROUP statements). Third, the REFMODEL statement (REFMODEL 1 / AllNewParms;), as a referencing tool, is used to build the model specification within the scope of the second MODEL statement (model 2 / group=2;), and it tells SAS that the pattern of fixed and free parameters in model specification 2 is equivalent to model specification 1. The AllNewParms option, standing for All New Parameters, means that the parameter names are totally different from those in Group 1, which allows parameters to vary between groups. By default, SAS adds suffix, _mdl2, to the corresponding parameter names in Group 1 so as to create unique names in Group 2. With unique names in the combined groups, there is no betweengroup constraint placed.

Starting Table 3, some modifications in the model for Group 2 are needed because some constraints are placed, and some constraints are removed. In Table 3 (Step 2a), constraints on all factor loadings are placed across groups, and other parameters are freely estimated in each group. This new model in Group 2 is specified by integrating some replacements to the referencing model referred to by the REFMODEL statement. In other words, the new entries in the PVAR and PCOV statements, nested within the scope of the REFMODEL statement, designate new parameter names in Group 2 by replacing those in previous PVAR and PCOV statements in Group 1; while the absence of new entries for the rest parameters implies between-group constraints. In Table 3, new parameter names are assigned in Group 2 by adding prefix, g2_, to the old parameter names, that is, the variances of F1, F2, V1, V2, ..., and V6 are g2_d1, g2_d2, g2_e1, g2_e2, ..., and g2_e6, respectively, and the covariance of F1 and F2 is g2_d12. The rest of parameters in Group 2 keep the same names as those in Group 1. By assigning the same parameter names across groups, constraints are placed. Also new in Table 3 is the MODIFICATION option in the PROC TCALIS statement. It requests the MI/LM tests on the constraints imposed on parameters in Group 1 and Group 2.

When we remove certain factor loadings, additional PATH statement is needed to replace the old path entries. In Table 4 (Step 2b, Substep 1), the second PATH statement, within the scope of the REFMODEL statement, specifies a new path entry, which tells SAS the loading from F2 to V3 is g2_b3 in Group 2. By designating a new name to the factor loading in Group 2, the constraint on loadings for V3 is removed. Similarly, in Table 4 (Step 2b, Substep 2 and Substep 3), two/three new path entries are specified to remove additional factor loadings in Group 2.

The use of the REFMODEL statement is not mandatory, if the user chooses to write out the complete model specification in Group 2. That is, users can retype all entries in Group 1 assigning the same, or different, parameter names in Group 2, depending on whether the between-group constraints are imposed or not. Usually, using the REFMODEL statement can save user's efforts in programming when model specification is very complicated.

When mean structures are introduced, the multigroup analyses become more complicated. In the MEAN statement, intercepts for measures and factors are specified. One or multiple variables can be listed on the left side of an equality sign, and correspondingly, one or multiple fixed values and/or unique parameter names need to be listed on the right side of an equality sign. Each specification is separated by a comma. For example, in Table 5, intercepts for factors are specified in Group 1 and Group 2, separately. That is, in Group 1, intercepts for F1 and F2 are set to 0, and intercepts for V1, V2, V4, and V5 are named as aV1, aV2, aV4, and aV5; whereas in Group 2, intercepts for F1, F2, V1, V2, V4, and V5 are freely estimated by assigning different parameter names: g2_aF1, g2_aF2, g2_aV1, g2_aV2, g2_aV4, and g2_aV5. In Table 6 through Table 8, different modifications are used in the MEAN statement accordingly due to the different between-group constraints placed on intercepts.

Table 2. Step 1—Evaluainge same factor model in each of the groups and combined groups

```
*** Group 1 ***;
proc tcalis data=group1 cov method=ml nobs=250;
fitindex NOINDEXTYPE on(only) = [CHISQ DF PROBCHI SRMSR RMSEA];
  path
    F1 -> V1 b1,
    F1 -> V2
              b2,
    F1 -> V3 b3,
    F2 -> V4 b4,
    F2 -> V5 b5,
    F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
    V1-V6 = e1-e6;
  pcov
    F1 F2 = d12;
run;
*** Group 2 ***;
proc tcalis data=group2 cov method=ml nobs=150;
fitindex NOINDEXTYPE on (only) = [CHISQ DF PROBCHI SRMSR RMSEA];
  path
    F1 -> V1 b1,
    F1 \rightarrow V2 b2,
    F1 \rightarrow V3 b3
    F2 \rightarrow V4 b4
    F2 \rightarrow V5 b5
    F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
    V1-V6 = e1-e6;
  pcov
    F1 F2 = d12;
run;
*** Combined groups ***;
proc tcalis method=ml;
fitindex NOINDEXTYPE on(only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
  path
```

```
F1 -> V1 b1,
    F1 -> V2 b2,
    F1 -> V3 b3,
    F2 \rightarrow V4 b4
    F2 \rightarrow V5 b5,
    F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
    V1-V6 = e1-e6;
  pcov
    F1 F2 = d12;
model 2 / group=2;
  refmodel 1 / AllNewParms;
run;
```

Table 3. Step 2a—Evaluating between-group equivalence of factor loadings

```
proc tcalis method=ml modification;
fitindex NOINDEXTYPE on (only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
  path
    F1 -> V1 b1,
    F1 \rightarrow V2 b2
    F1 -> V3
              b3,
    F2 -> V4 b4,
    F2 -> V5 b5,
    F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
    V1-V6 = e1-e6;
    F1 F2 = d12;
model 2 / group=2;
  refmodel 1;
  pvar
    F1 = g2 d1,
    F2 = g2 d2,
    V1-V6 = g2_e1-g2_e6;
  pcov
    F1 F2 = g2 d12;
```

Table 4. Step 2b—Evaluating between-group equivalence of individual loadings

```
*** Step 2b, Substep 1: remove the constraint on loadings for V3 ***;
proc tcalis method=ml modification;
fitindex NOINDEXTYPE on (only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
  path
    F1 -> V1 b1,
    F1 \rightarrow V2 b2
    F1 -> V3
              b3,
```

```
F2 -> V4 b4,
    F2 -> V5 b5,
    F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
    V1-V6 = e1-e6;
  pcov
    F1 F2 = d12;
model 2 / group=2;
  refmodel 1;
  path
   F1 -> V3 g2_b3;
  pvar
   F1 = g2_d1,
    F2 = g2 d2,
    V1-V6 = g2_e1-g2_e6;
    F1 F2 = g2 d12;
run;
*** Step 2b, Substep 2: remove the constraint on loadings for V6 ***;
proc tcalis method=ml modification;
fitindex NOINDEXTYPE on(only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
  path
    F1 -> V1 b1,
    F1 -> V2 b2,
    F1 -> V3 b3,
    F2 -> V4 b4,
    F2 \rightarrow V5 b5,
    F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
    V1-V6 = e1-e6;
  pcov
    F1 F2 = d12;
model 2 / group=2;
  refmodel 1;
  path
    F1 \rightarrow V3 g2_b3,
    F2 -> V6 g2 b6;
  pvar
    F1 = g2 d1,
    F2 = g2 d2,
    V1-V6 = g2_e1-g2_e6;
  pcov
    F1 F2 = g2_d12;
run;
```

```
*** Step 2b, Substep 3: remove the constraint on loadings for V1 ***;
proc tcalis method=ml modification;
fitindex NOINDEXTYPE on (only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
  path
   F1 -> V1 b1,
    F1 \rightarrow V2 b2
   F1 -> V3 b3,
   F2 -> V4 b4,
   F2 -> V5 b5,
   F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
    V1-V6 = e1-e6;
  pcov
    F1 F2 = d12;
model 2 / group=2;
  refmodel 1;
  path
   F1 -> V1 g2_b1,
    F1 -> V3 g2 b3,
    F2 -> V6 g2 b6;
  pvar
   F1 = g2 d1,
    F2 = g2 d2,
    V1-V6 = g2 e1-g2 e6;
    F1 F2 = g2_d12;
run;
```

Table 5. Step 3a—Evaluating between-group equivalence of intercepts for measures

```
proc tcalis method=ml modification maxiter=300 maxfunc=300;
fitindex NOINDEXTYPE on(only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
  path
   F1 -> V1 b1,
    F1 -> V2 b2,
   F1 -> V3 b3,
   F2 -> V4 b4,
   F2 \rightarrow V5 b5,
    F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
    V1-V6 = e1-e6;
  pcov
    F1 F2 = d12;
    F1 F2 = 0 0,
    V1-V6 = aV1-aV6;
model 2 / group=2;
```

```
refmodel 1;

path
    F1 -> V3     g2_b3,
    F2 -> V6     g2_b6;

pvar
    F1 = g2_d1,
    F2 = g2_d2,
    V1-V6 = g2_e1-g2_e6;

pcov
    F1 F2 = g2_d12;

mean
    F1 F2 = g2_aF1    g2_aF2,
    V3 V6 = g2_aV3    g2_aV6;

run;
```

```
Table 6. Step 3b—Evaluating between-group equivalence of individual intercepts for measures
*** Step 3b, Substep 1: remove the constraint on intercepts for V1 ***;
proc tcalis method=ml modification maxiter=300 maxfunc=300;
fitindex NOINDEXTYPE on (only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
  path
    F1 -> V1 b1,
    F1 -> V2 b2,
    F1 -> V3 b3,
    F2 -> V4 b4,
    F2 -> V5 b5,
    F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0
    V1-V6 = e1-e6;
    F1 F2 = d12;
  mean
    F1 F2 = 0 0,
    V1-V6 = aV1-aV6;
model 2 / group=2;
  refmodel 1;
  path
   F1 -> V3 g2 b3,
    F2 -> V6 g2 b6;
  pvar
    F1 = g2 d1,
    F2 = g2 d2,
    V1-V6 = g2_e1-g2_e6;
  pcov
    F1 F2 = g2_d12;
  mean
    F1 F2 = g2 aF1 g2 aF2,
    V1 V3 V6 = g2_aV1 g2_aV3 g2_aV6;
run;
*** Step 3b, Substep 2: remove the constraint on intercepts for V4 ***;
```

```
proc tcalis method=ml modification maxiter=300 maxfunc=300;
fitindex NOINDEXTYPE on(only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
 path
   F1 -> V1 b1,
   F1 -> V2 b2,
   F1 -> V3
             b3,
   F2 -> V4 b4,
   F2 -> V5 b5,
   F2 -> V6 b6;
  pvar
   F1 = 1.0,
   F2 = 1.0,
   V1-V6 = e1-e6;
 pcov
   F1 F2 = d12;
 mean
    F1 F2 = 0 0,
   V1-V6 = aV1-aV6;
model 2 / group=2;
  refmodel 1;
 path
   F1 -> V3 g2_b3,
   F2 -> V6 g2 b6;
   F1 = g2_d1,
   F2 = g2 d2,
   V1-V6 = g2 e1-g2 e6;
   F1 F2 = g2_d12;
 mean
   F1 F2 = g2_aF1 g2_aF2,
   V1 V3 V4 V6 = g2_aV1 g2_aV3 g2_V4 g2_aV6;
```

Table 7. Step 4a—Evaluating between-group equivalence of intercepts for factors

```
proc tcalis method=ml modification maxiter=300 maxfunc=300;
fitindex NOINDEXTYPE on(only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
  path
   F1 -> V1 b1,
    F1 \rightarrow V2 b2
    F1 -> V3 b3,
   F2 -> V4 b4,
   F2 -> V5 b5,
    F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
   V1-V6 = e1-e6;
  pcov
    F1 F2 = d12;
  mean
    F1 F2 = 0 0,
```

```
V1-V6 = aV1-aV6;
model 2 / group=2;
refmodel 1;

path
   F1 -> V3    g2_b3,
   F2 -> V6    g2_b6;
pvar
   F1 = g2_d1,
   F2 = g2_d2,
   V1-V6 = g2_e1-g2_e6;
pcov
   F1 F2 = g2_d12;
mean
   V1 V3 V4 V6 = g2_aV1 g2_aV3 g2_V4 g2_aV6;
run;
```

```
Table 8. Step 4b—Evaluating between-group equivalence of individual intercepts for factors
*** Step 4b, Substep 1: remove the constraint on intercepts for F1 ***;
proc tcalis method=ml modification maxiter=300 maxfunc=300;
fitindex NOINDEXTYPE on(only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
  path
    F1 -> V1 b1,
    F1 -> V2 b2,
    F1 -> V3 b3,
    F2 -> V4 b4,
    F2 -> V5 b5,
   F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
    V1-V6 = e1-e6;
  pcov
    F1 F2 = d12;
  mean
    F1 F2 = 0 0,
    V1-V6 = aV1-aV6;
model 2 / group=2;
  refmodel 1;
  path
   F1 -> V3 q2 b3,
    F2 -> V6 g2 b6;
  pvar
    F1 = g2 d1,
    F2 = g2 d2,
    V1-V6 = g2_e1-g2_e6;
  pcov
    F1 F2 = g2 d12;
  mean
    F2 = g2 aF2,
    V1 V3 V4 V6 = g2_aV1 g2_aV3 g2_V4 g2_aV6;
run;
```

```
*** Step 4b, Substep 2: remove the constraint on intercepts for F2 ***;
proc tcalis method=ml modification maxiter=300 maxfunc=300;
fitindex NOINDEXTYPE on (only) = [CHISQ DF PROBCHI SRMSR RMSEA];
group 1 / data=group1 nobs=250;
group 2 / data=group2 nobs=150;
model 1 / group=1;
  path
    F1 -> V1 b1,
    F1 -> V2
              b2,
    F1 \rightarrow V3 b3
   F2 -> V4 b4,
    F2 -> V5 b5,
    F2 -> V6 b6;
  pvar
    F1 = 1.0,
    F2 = 1.0,
    V1-V6 = e1-e6;
  pcov
    F1 F2 = d12;
  mean
    F1 F2 = 0 0.
    V1-V6 = aV1-aV6;
model 2 / group=2;
  refmodel 1;
  path
    F1 -> V3 g2 b3,
    F2 -> V6 q2 b6;
    F1 = q2 d1,
    F2 = g2 d2
    V1-V6 = g2_e1-g2_e6;
  pcov
    F1 F2 = g2 d12;
  mean
    F1 = q2 aF1,
    V1 V3 V4 V6 = g2 aV1 g2 aV3 g2 V4 g2 aV6;
```

As noted before, when mean structures are involved, multigroup SEM gets more complicated. As a result, nonconvergence may occur in the estimation process. One possible solution is to increase both the maximum number of iteration and function call. The MAXITER= and MAXFUNC= options in the PROC TCALIS statement allows the user to set different numbers of iteration and function call (Table 5 – Table 8). An alternative is to provide starting values. In a PATH model, adding a value in paired parentheses after the parameter name tells SAS to use this value as the starting value. For example, the following PATH entry,

means that PROC TCALIS will use 0.5 as the starting value to estimate the parameter b1.

The use of the FITINDEX statement is optional. In this example, the NOINDEXTYPE option disables the display of index types in the fit summary table, and the ON(ONLY)= option turns on the printing of chi-square, degree of freedom, p-value, SRMR, and RMSEA. If the FITINDEX statement is omitted, a complete fit summary table is displayed.

RESULTS

Model fit indices from all steps are collected in Table 9. As noted by Thompson and Green (2006, p.156), fit indices produced by EQS were reported in the chapter. For single-group models, the values of SRMR and RMSEA are identical between EQS and SAS 9.2. However, for multigroup models, different values of SRMR are reported by SAS

compared to EQS. Some of the differences are not trivial (Step 3a, Step 4a, and Step 4b Substep 2). In PROC TCALIS, the SRMR for each group is calculated first, and then, the group weight, $a_r = \frac{N_r - 1}{N - k}$, is used to calculate the overall SRMR, that is,

Overall SRMR =
$$\sqrt{\sum_{r=1}^{k} a_r SRMR_r^2}$$

where SRMR_r is the SRMR for Group r, N_r is the sample size of Group r, N is the total sample size, and k is the number of groups. If the FITINDEX statement is omitted, the user can find the SRMR for each group in the complete fit summary table. In contrast, there are two reasons that EQS gives different SRMRs. First, when there is mean structure, EQS bases the calculation on covariances only, and ignores mean residuals; and second, EQS calculates the overall SRMR in a direct way, that is, the squared residuals in each group are added up together and then divided by kp(p+1), where k is the number of groups, and p is the number of observed variables in a group (David Sookne, personal communication, August, 2010). The discrepancies in the values of RMSEA result from the adjustment for multigroup model proposed by Steiger (1998), That is, an averaged degrees of freedom should be employed in the calculation of the RMSEA. As Thompson and Green (2006, p.158) noted, EQS had not yet incorporated Steiger's correction at the time of the writing of the chapter, but it does in EQS 6.1. In this example, after multiplying $\sqrt{2}$ to the RMSEA values reported by EQS, they are the same as those reported by SAS. As a cross-validation, we also fit all the example models in Mplus 5.21, and the values of SRMR and RMSEA agree with SAS 9.2 perfectly. Many other fit indices are available in PROC TCALIS, such as the comparative fit index and the non-normal fit index (also known as Tucker-Lewis index). Technical details of these fit indices in PROC TCALIS can be found from the SAS/STAT user's quide (SAS Institute, 2008, pp.6892-6905).

Table 9. Selected model fit indices results from SAS 9.2 and the book chapter

	χ^2			SRMR		RMSEA	
	degrees of	book	SAS 9.2	book	SAS 9.2	book	SAS
	freedom	chapter		chapter		chapter	9.2
Step 1: Goup 1	8	8.02	8.02	.031	.031	.003	.003
Step 1: Goup 2	8	9.77	9.77	.045	.045	.039	.039
Step 1: Goup 1 & 2 combined	16	17.79	17.79	.039	.037	.017	.024
Step 2a: place constraints on all loadings	20	26.07	26.07	.048	.048	.028	.039
Step 2b, Substep 1: remove constraint on loading for V3	19	22.04	22.04	.043	.042	.020	.028
Step 2b, Substep 2: remove constraint on loading for V6	18	19.16	19.16	.041	.040	.013	.018
Step 2b, Substep 3: remove constraint on loading for V1	17	17.86	17.86	.039	.037	.011	.016
Step 3a: place constraints on all intercepts except for V3 and V6	20	54.44	54.44	.053	.060	.066	.093
Step 3b, Substep 1: remove constraint on intercepts for V1	19	29.43	29.43	.047	.044	.037	.053
Step 3b, Substep 2: remove constraint on intercepts for V4	18	19.16	19.16	.041	.035	.013	.018
Step 4a: place constraints on intercept for both factors	20	56.39	56.39	.061	.092	.068	.096
Step 4b, Substep 1: remove constraint on intercepts for F1	19	21.28	21.28	.041	.039	.017	.025
Step 4b, Substep 2: remove constraint on intercepts for F2	19	56.17	56.17	.060	.090	.070	.099

Note. SRMR: standardized root mean square residual. RMSEA: root mean square error of approximation.

Parameter estimates reported by PROC TCALIS are identical to those obtained by EQS (Table 5.5, p.150), and they are omitted here.

COMMENTS AND CONCLUSIONS

In addition to the tests of difference in factor means illustrated in this example, invariance can also be tested at more stringent levels, such as factor variances and covariances, and possibly error variances. Though the length of this article restrained complete illustrations of all of them, any invariance test can be easily carried out as long as one

follows the general principle such that, for a particular parameter in one group, identical names in other groups imply constraints while different names designate freely estimated parameters.

Other modeling languages can be used to specify the models in this illustration. Users knowledgeable of EQS and LISREL will find it easy to specify models by using the LINEQS and LISMOD language, separately. In fact, the layout forms of the output by using different modeling languages (PATH, LINEQS, and LISMOD) are similar to those generated by specialized SEM software packages (Mplus, EQS, and LISREL).

In addition to normal-theory maximum likelihood method, other estimation methods are also available. For example, the METHOD=WLS requests weighted least squares method. The user needs to provide a nonsingular weight matrix in the INWGT= option in PROC TCALIS. If no weight matrix is provided, the WLS method uses the inverse matrix of estimated asymptotic covariances of the sample covariance or correlation matrix as the weight matrix. In this case, the WLS method is equivalent to Browne's asymptotically distribution-free estimation (Browne, 1982, 1984).

Though the TCALIS procedure in SAS 9.2 is greatly improved from the old CALIS procedure, there are still some limitations. For example, PROC TCALIS is not capable of analyzing ordered-categorical data, or data with nested structures (multilevel SEM).

Historically, SEM models specified in SAS PROC CALIS are very limited in scope. In this article, the TCALIS procedure is illustrated for multigroup SEM analyses. The fit indices (i.e., chi-square, SRMR, RMSEA) and parameter estimates are validated.

REFERENCES

- Arbuckle, J. L., (2003). AMOS (Version 5.0) [Computer software]. Chicago: Smallwaters.
- Bentler, P. M. & Wu, E. J. C. (2002). EQS 6 for Windows guide. Encino, CA: Multivariate Software.
- Bollen, K. A. (1989). Structural equations with latent variables. New York: Wiley.
- Browne, M. W. (1982). Covariance structures. In D. M. Hawkins (Ed.), Topics in applied multivariate analysis (pp. 72-141). Cambridge: Cambridge University Press.
- Browne, M. W. (1984). Asymptotically distribution-free methods for the analysis of covariance structures. British Journal of Mathematical and Statistical Psychology, 37, 62-83.
- Byrne, B. M. (2004). Testing for multigroup invariance using AMOS graphics: A road less traveled. Structural Equation Modeling, 11(2), 272-300.
- Byrne, B. M., Shavelson, R. J., & Muthén, B. (1989). Testing for the equivalence of factor covariance and mean structures: The issue of partial measurement invariance. Psychological Bulletin, 105, 456-466.
- Chen, F. F. (2007). Sensitivity of goodness of fit indexes to lack of measurement invariance. Structural Equation Modeling, 14(3), 464-504.
- Cheung, G. W. & Rensvold, R. B. (2002). Evaluating goodness-of-fit indexes for testing measurement invariance. Structural Equation Modeling, 9(2), 233-255.
- Chou, C.-P. & Bentler, P. M. (1990). Model modification in covariance structure modeling: A comparison among likelihood ratio, Lagrange multiplier, and Wald tests. Multivariate Behavioral Research, 25, 115-136.
- Fan, X. & Fan, X. (2005). Using SAS for Monte Carlo simulation research in SEM. Structural Equation Modeling, 12(2), 299-333.
- Fan, X., Felsövályi, Á., Sivo, S. A., & Keenan, S. (2003). SAS for Monte Carlo studies: A guide for quantitative researchers. Cary, NC: SAS Institute.
- Fan, X. & Sivo, S. A., (2005). Sensitivity of fit indexes to misspecified structural or measurement model components: Rationale of two-index strategy revisited. Structural Equation Modeling, 12(3), 343-367.
- Fan, X. & Sivo, S. A., (2009). Using ∆goodness-of-fit indexes in assessing mean structure invariance. Structural Equation Modeling, 16, 54-69.
- Jones-Farmer, L. A., Pitts J. P., & Rainer, R. K. (2008). A note on multigroup comparisons using SAS PROC CALIS. Structural Equation Modeling, 15, 154-173.
- Jöreskog, K. G. & Sörbom, D. (1996). LISREL 8: User's reference guide. Chicago: Scientific Software International.

- Little, T. D. (1997). Mean and covariance structures (MACS) analyses of cross-cultural data: Practical and theoretical issues. Multivariate Behavioral Research, 32, 53-76.
- Marcoulides, G. A. & Hershberger, S. L. (1997). Multivariate statistical methods: A first course. Mahwah, NJ: Lawrence Erlbaum Associates, Inc.
- Marsh, H. W., Muthén, B. O., Asparouhov, T., Lüdtke, O., Robitzsch, A., Morin, A. J. S., & Trautwein U. (2009). Exploratory structural equation modeling, integrating CFA and EFA: Application to students' evaluations of university teaching. Structural Equation Modeling, 16(3), 439-476.
- McDonald, R. P. & Ho, M.-H. R. (2002). Principles and practice in reporting structural equation analyses. Psychological Methods, 7(1), 64-82.
- Meredith, W. (1993). Measurement invariance, factor analysis and factorial invariance. Psychometrika, 58, 525-543.
- Muthén, L. K. & Muthén, B. O. (1998-2007). Mplus user's guide. Los Angeles, CA: Muthén & Muthén.
- Neale, M.C., Boker, S. M., Xie, G., & Maes, H. H. (2003). Mx: statistical modeling. VCU Box 900126, Richmond, VA 23298: Department of Psychiatry. 6th Edition.
- SAS Institute Inc. (2008) SAS/STAT® 9.2 User's Guide. Cary, NC: SAS Institute Inc.
- Steiger, J. H. (1998). A note on multiple sample extensions of the RMSEA fit index. Structural Equation Modeling, 5(4), 411-419.
- Thompson M. S. & Green, S. B. (2006). Evaluating between-group differences in latent variable means. In G. R. Hancock & R. O. Mueller (Eds.), Structural Equation Modeling: A second course (pp. 119-169). Greenwich, CT: Information Age Publishing, Inc.
- Vandenberg, R. J. & Lance, C. E. (2000). A review and synthesis of the measurement invariance literature: Suggestions, practices, and recommendations for organizational research. Organizational Research Methods, 3, 4-69.
- Yang, Y. & Green, S. B., (2010). A note on structural equation modeling estimates of reliability. Structural Equation Modeling, 17(1), 66-81.

ACKNOWLEDGMENTS

This work is supported by the University of Kansas' Life Span Institute/Research Design & Analysis unit (LSI/RDA) and the Center for Educational Testing & Evaluation (CETE).

RECOMMENDED READING

This paper was peer reviewed and revised, and it is published in the British Journal of Mathematical and Statistical Psychology. It can be retrieved from http://dx.doi.org/10.1111/j.2044-8317.2010.02012.x.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Fei Gu Psychology and Research in Education University of Kansas 1122 W Campus Rd. Lawrence, KS 66045 785-864-9701 fgu@ku.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.